Projet Génie Logiciel

Étape C

Génération de code

Projet GL

Ensimag Grenoble INP

2 janvier 2023



2 janvier 2023

Sommaire

- Génération de code impératif simple
- Que de la code orienté objet



Sommaire de cette section

- 🕕 Génération de code impératif simple
 - Organisation de la mémoire à l'exécution dans IMA
 - Exercice 1
 - Documentation
 - Algorithmes et implantation de la génération de code



Organisation de la mémoire à l'exécution dans IMA

Tas

Pile
← SP

Données statiques
← GB

Code du programme

données statiques variables globales, littéraux du programme, tables

littéraux du programme, table des méthodes, et valeurs temporaires

pile variables locales, paramètres de méthodes

tas objets alloués dynamiquement (pas utilisé dans le langage sans-objet)

2 janvier 2023

GB (« Global Base ») Registre particulier qui pointe sur le fond de pile

SP (« Stack Pointer ») Pointe sur la dernière case occupée de la pile



Analogie/Différences avec l'assembleur x86 (Pentium)

- zone statique IMA = sections .data, .rodata et .bss en ELF x86.
- SP pprox %esp (%rsp en 64 bits)
- GB n'a pas d'équivalent en x86



Principes de la génération de code pour expressions

- Évaluation de gauche à droite (sémantique de Deca)
- Calcul du résultat dans un registre banalisé $\geq \mathbf{R}2$, en utilisant d'autres registres pour sauvegarde des résultats de sous-expressions.
- Registres R0 et R1 = registres scratch (modifiables par les appels de méthodes), pas utilisés pour sauvegarder le résultat de sous-expressions.
- RMAX autorisé: X-1 si option -r X, ou 15 sinon.
- Si plus de registre disponible, utiliser PUSH et POP pour sauvegarde des résultats sur la pile (c-à-d. des "temporaires").
 - NB : attention à prise en compte dans calcul du TSTO (cf. plus loin).

Principes de la génération de code pour expressions

- Évaluation de gauche à droite (sémantique de Deca)
- Calcul du résultat dans un registre banalisé $\geq R2$, en utilisant d'autres registres pour sauvegarde des résultats de sous-expressions.
- Registres R0 et R1 = registres scratch (modifiables par les appels de méthodes), pas utilisés pour sauvegarder le résultat de sous-expressions.
- RMAX autorisé : X-1 si option -r X, ou 15 sinon.
- Si plus de registre disponible, utiliser PUSH et POP pour sauvegarde des résultats sur la pile (c-à-d. des "temporaires").

NB: attention à prise en compte dans calcul du TSTO (cf. plus loin).

Exemple de l'affectation

```
< code Exp(Assign[Ident\uparrowsymb\ e], n) > 
:= < code Exp(e, n) > // calcul\ de\ e\ dans\ Rn\ (avec\ n \in 2..MAX)
STORE Rn, @symb
```



Opérande d'une expression atomique

```
\begin{array}{l} <\!\!\mathrm{d}\,\mathrm{val}\big(\,\underline{\mathtt{IntLiteral}}\!\!\uparrow\!\!n\big)\!\!>\; :=\; \#\,n \\ <\!\!\mathrm{d}\,\mathrm{val}\big(\,\underline{\mathtt{Identifier}}\!\!\uparrow\!\!symb\,\big)\!\!>\; :=\; @\,symb \\ <\!\!\mathrm{d}\,\mathrm{val}\big(\,\underline{-}\big[\,\underline{-}\,\underline{-}\big]\,\big)\!\!>\; :=\; \bot \end{array}
```

Opérande d'une expression atomique

Mnémonique d'un opérateur binaire

Opérande d'une expression atomique

Mnémonique d'un opérateur binaire

Code pour calculer e dans Rn (utilisant uniquement R0 et Rn ... RMAX)

```
 \begin{aligned} &<\operatorname{codeExp}(\ e, \ n)> \\ & \operatorname{avec} \ &<\operatorname{dval}(\ e)> \neq \bot \\ &:= \ \mathsf{LOAD} \ &<\operatorname{dval}(\ e)>, \ \mathsf{R} n \end{aligned}   <& \operatorname{codeExp}(\ op[\ e1 \ e2]\ , n)> \\ & \operatorname{avec} \ &<\operatorname{dval}(\ e2)> \neq \bot \\ &:= \ &<\operatorname{codeExp}(\ e1\ , n)> \\ & <\operatorname{mnemo}(\ op)> \ &<\operatorname{dval}(\ e2)>, \ \mathsf{R} n \end{aligned}
```

Opérande d'une expression atomique

Mnémonique d'un opérateur binaire

Code pour calculer e dans Rn (utilisant uniquement R0 et Rn ... RMAX)

Opérande d'une expression atomique

Mnémonique d'un opérateur binaire

Code pour calculer e dans Rn (utilisant uniquement R0 et Rn ... RMAX)

```
 \begin{aligned} &<\!\!\operatorname{codeExp}(\mathit{op[e1\ e2]},\mathit{n})\!> \mathtt{avec} <\!\!\operatorname{dval}(\mathit{e2})\!> = \perp \mathtt{et} \mathit{n} <\!\!\mathsf{MAX} \\ &:= <\!\!\operatorname{codeExp}(\mathit{e1},\mathit{n})\!> \\ &<\!\!\operatorname{codeExp}(\mathit{e2},\mathit{n} + \!\!1)\!> \\ &<\!\!\operatorname{mnemo}(\mathit{op})\!> \mathsf{R} \mathit{n} + \!\!1, \mathsf{R} \mathit{n} \end{aligned}
```

Sommaire de cette section

- Génération de code impératif simple
 - Organisation de la mémoire à l'exécution dans IMA
 - Exercice 1
 - Documentation
 - Algorithmes et implantation de la génération de code



```
On considère le programme suivant :
{
   int x = 1;
   int y = 2;
   int z;
   z = 2 * x - 3 * y;
```

Question




Dessiner la pile

Question




Écrire l'assembleur généré



Rem : Table de méthodes de Object pas nécessaire sans objet.


Pile :



• Code généré (RMAX≥R3) :

```
; int x = 1
LOAD #1, R2
STORE R2, 3(GB)
; int y = 2
LOAD #2, R2
STORE R2, 4(GB)
z = 2 * x - 3 * y
LOAD #2, R2
MUL 3(GB), R2
LOAD #3, R3
MUL 4(GB), R3
SUB R3, R2
STORE R2, 5(GB)
```

Pile :



Code généré (RMAX=R2) :

```
; int x = 1
LOAD #1, R2
STORE R2, 3(GB)
; int y = 2
LOAD #2, R2
STORE R2, 4(GB)
z = 2 * x - 3 * v
LOAD #2, R2
MUL 3(GB), R2
PUSH R2 ; sauvegarde
LOAD #3, R2
MUL 4(GB), R2
LOAD R2, RO
POP R2 ; restauration
SUB RO, R2
STORE R2, 5(GB)
```



Sommaire de cette section

- 🕕 Génération de code impératif simple
 - Organisation de la mémoire à l'exécution dans IMA
 - Exercice 1
 - Documentation
 - Algorithmes et implantation de la génération de code



Documentation

- Sémantique de Deca : II-[Semantique]
- Machine abstraite : II-[MachineAbstraite]
- Conventions de liaison : II-[ConventionsLiaison]
- Algorithmes de génération de code : IV-[Gencode]
- Outil fourni : ima (interprète de la machine abstraite), plus un metteur au point (voir IV-[Ima])
- À faire : I-[Consignes] (étape C)

Sommaire de cette section

- Génération de code impératif simple
 - Organisation de la mémoire à l'exécution dans IMA
 - Exercice 1
 - Documentation
 - Algorithmes et implantation de la génération de code



Algorithmes et implantation de la génération de code

IV-[Gencode]



Génération de code impératif simple

Algorithmes et implantation de la génération de code

IV-[Gencode]

- Génération de code en deux passes :
 - ▶ Passe 1 : construction de la table des méthodes de chaque classe
 - ★ Pas utile pour le langage sans-objet.
 - ★ cf. IV-[Exemple]
 - Passe 2 : génération de code pour
 - ★ le programme principal
 - ★ chaque classe : initialisation des objets et codage des méthodes
 - ★ messages d'erreurs pour les erreurs à l'exécution
- Parcours d'arbre en utilisant le patron « interprète », basé sur la grammaire d'arbres.



Sommaire

- Génération de code impératif simple
- Génération de code orienté objet




Sommaire de cette section

- Que Génération de code orienté objet
 - Organisation de la mémoire
 - Exercice 2




Bloc d'activation d'un appel de procédure



- SP Pointeur de Pile, qui pointe sur le sommet de pile. (analogie avec l'assembleur x86 : SP ≈ %esp, ou %rip en 64 bits)
- LB base locale, qui permet d'accéder aux paramètres, variables locales et temporaires dans la pile. (analogie avec l'assembleur x86 : LB ≈ %ebp, ou %rbp en 64 bits)


Table des méthodes

```
class A {
                           Ω
   int x;
                           méthodes de
   int y;
   void m() { }
   void p() { }
}
                           Table des
class B extends A {
   int x;
   int z;
   void p() { }
   void q() { }
```



Codage d'un objet

Objet o instance de classe B :



Sommaire de cette section

- 2 Génération de code orienté objet
 - Organisation de la mémoire
 - Exercice 2



Exercice 2 : programme d'exemple

```
class Point2D {
   int x; // Abscisse
   int y; // Ordonnee
   // Deplace ce point
   // de a en diagonale.
   void diag(int a) {
      x = x + a;
      y = y + a;
class Point3D extends Point2D {
   int z; // Hauteur
   // On redefinit la methode diag,
   // pour tenir compte de z
   void diag(int a) {
      x = x + a:
      y = y + a;
      z = z + a:
```

```
Point2D p1, p2;
   Point3D p3;
   p1 = new Point2D();
   p1.diag(1);
   p3 = new Point3D();
   p3.diag(2);
   p2 = p3;
   p2.diag(3);
   println("p3.z = ", p3.z);
}
```

- Qu'affiche le programme?
- Dessiner l'état de la pile et du tas
- Ecrire le code généré pour ce programme Deca




Exercice 2(a) : Sémantique du programme

```
Point2D p1, p2;
Point3D p3;
p1 = new Point2D(); // p1 initialise a zero :
                    // p1.x = 0; p1.y = 0
                    // p1.x = 1; p1.y = 1;
p1.diag(1);
p3 = new Point3D(); // p3 initialise a zero :
                    // p3.x = 0; p3.y = 0; p3.z = 0
                    // p3.x = 2; p3.y = 2; p3.z = 2
p3.diag(2);
p2 = p3; // p2 et p3 representent le meme point
p2.diag(3); // Liaison dynamique : Appel de Point3D.diag
             // p2.x = 5; p2.y = 5; p2.z = 5
println("p3.z = ", p3.z); // p3.z = 5
```

}

Exercice 2(b) : état de la pile et du tas



Construction des tables de méthodes

```
ADDSP #11 (8 pour tables des methodes + 3 variables globales)
; Construction de la table des methodes de Object
    LOAD #null, RO
    STORE RO. 1 (GB)
    LOAD code.Object.equals, RO
    STORE RO. 2 (GB)
: Construction de la table des methodes de Point2D
    LEA 1 (GB), RO
    STORE RO. 3 (GB)
    LOAD code.Object.equals, RO
    STORE RO, 4 (GB)
    LOAD code.Point2D.diag, RO
    STORE RO, 5 (GB)
: Construction de la table des methodes de Point3D
    LEA 3 (GB), RO
    STORE RO, 6 (GB)
    LOAD code.Object.equals, RO
    STORE RO, 7 (GB)
    LOAD code.Point3D.diag, RO
    STORE RO. 8 (GB)
```

2 janvier 2023

Initialisation à 0 des champs de Point2D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)

2 janvier 2023

Initialisation à 0 des champs de Point2D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)

```
; Initialisation des
; champs de Point2D
init.Point2D:
LOAD #0, RO
LOAD -2(LB), R1
STORE RO, 1(R1)
LOAD #0, RO
LOAD -2(LB), R1
STORE RO, 2(R1)
RTS
```

Code de la méthode diag de Point2D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)
-3(LB)	paramètre a

Code de la méthode diag de Point2D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)
-3(LB)	paramètre a

```
; Code de la methode diag
; dans la classe Point2D
code.Point2D.diag :
; Sauvegarde des registres
    PUSH R2
    PUSH R3
x = x + a
    LOAD -2 (LB), R2
    LOAD - 2 (LB), R3
    LOAD 1 (R3), R3
     ADD -3 (LB), R3
    STORE R3. 1 (R2)
; y = y + a
    LOAD -2 (LB), R2
    LOAD -2 (LB), R3
    LOAD 2 (R3), R3
     ADD -3 (LB), R3
    STORE R3, 2 (R2)
; Restauration des registres
    POP R3
    POP R2
```

RTS

Initialisation à 0 des champs de Point3D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)

Initialisation à 0 des champs de Point3D

LB	Ancien LB
-1(LB)	Adresse de retour
-2(LB)	Adresse de l'objet (this)

```
Initialisation des
 champs de Point3D
init.Point3D :
     LOAD -2 (LB), RO
     PUSH RO
     BSR init.Point2D
     SUBSP #1
     LOAD #0, RO
     LOAD -2 (LB), R1
     STORE RO, 3 (R1)
     RTS
```

Code de la méthode diag de Point3D

```
; Code de la methode diag
 dans la classe Point3D
code.Point3D.diag :
; Sauvegarde des registres
     PUSH R2
     PUSH R3
 x = x + a
     LOAD -2 (LB), R2
     LOAD -2 (LB), R3
     LOAD 1 (R3), R3
     ADD -3 (LB), R3
     STORE R3, 1 (R2)
 y = y + a
     LOAD -2 (LB), R2
     LOAD -2 (LB), R3
     LOAD 2 (R3), R3
     ADD -3 (LB), R3
     STORE R3, 2 (R2)
```

```
; z = z + a

LOAD -2 (LB), R2

LOAD -2 (LB), R3

LOAD 3 (R3), R3


ADD -3 (LB), R3

STORE R3, 3 (R2)

POP R3

POP R2

RTS
```



2 janvier 2023



Code du programme principal (1)

; p1.diag(1); ADDSP #2

> BSR 2 (R2) SUBSP #2

> > 2 janvier 2023

```
; empile p1
                                            LOAD 9 (GB), R2
                                            STORE R2, O (SP)
; p1 = new Point2D();
                                            ; empile 1
; (allocation + initialisation)
                                            LOAD #1, R2
    NEW #3, R2
                                            STORE R2, -1 (SP)
    BOV tas_plein
                                            ; appel de méthode
    LEA 3 (GB), RO
                                            LOAD O (SP), R2
    STORE RO, O (R2)
                                            ; objet null dans
    PUSH R2
                                            ; appel de methode ?
    BSR init.Point2D
                                            CMP #null, R2
    POP R.2
                                            BEQ dereferencement_null
    STORE R2, 9 (GB)
                                            ; adresse de la
                                            ; méthode diag de p1.
                                            LOAD O (R2), R2
```

Code du programme principal (2)

```
p3 = new Point3D();
   NEW #4, R2
   BOV tas_plein
   LEA 6 (GB), RO
   STORE RO, O (R2)
   PUSH R2
   BSR init.Point3D
   POP R2
   STORE R2, 11 (GB)
```

```
; p3.diag(2);
     ADDSP #2
     ; empile p3
     LOAD 11 (GB), R2
     STORE R2, 0 (SP)
     ; empile 2
    LOAD #2, R2
     STORE R2, -1 (SP)
     ; appel de méthode
    LOAD O (SP), R2
     ; objet null
     ; dans appel de methode ?
     CMP #null, R2
     BEQ dereferencement_null
     ; adresse de la
     ; méthode diag de p3.
     LOAD 0 (R2), R2
     BSR 2 (R2)
     SUBSP #2
```

Code du programme principal (3)

```
; p2 = p3;
    LOAD 11 (GB), R2
    STORE R2, 10 (GB)
; p2.diag(3);
; Liaison dynamique :
; Appel de Point3D.diag
;
; Comme précédemment
```

```
; println("p3.z = ", p3.z);
; affiche : "p3.z = 5"
    WSTR "p3.z = "
     LOAD 11 (GB), R2
     ; objet null dans
     ; selection de champ ?
     CMP #null, R2
     BEQ dereferencement null
    LOAD 3 (R2), R2
    LOAD R2, R1
    WINT
     WNT.
     HAT.T
```

Problème du TSTO

Lignes à ajouter aux extraits de code ci-dessus pour tester les débordements de pile :

```
code.Point2D.diag :
   TST0 #2
   BOV pile_pleine
code.Point3D.diag :
   TST0 #2
   BOV pile_pleine
init.Point3D:
   ; pour pouvoir appeler
   : init.Point2D:
   TST0 #3
   BOV pile_pleine
```

Récapitulatif des erreurs à l'exécution

II [Semantique]

- erreurs de débordement :
 - pile
 - tas
 - arithmétique (sur flottants, inclus la division par 0.0)
- division entière par 0 et reste de la division entière par 0
- sortie de méthode sans passer par "return"
- conversion de type impossible
- déréférencement de null
- erreur de lecture (RINT et on ne tape pas un entier; RFLOAT et on ne tape pas un flottant)
- accès à des variables non initialisées (non traité : si on y accède ima donne un message d'erreur).

